

CHAPTER 22

The GPRINT Procedure

<i>Overview</i>	859
<i>Concepts</i>	860
<i>About External Text Files</i>	860
<i>Procedure Syntax</i>	860
<i>PROC GPRINT Statement</i>	860
<i>Examples</i>	865
<i>Example 1: Specifying Color Text</i>	865
<i>Example 2: Adjusting the Size of Characters</i>	868

Overview

The GPRINT procedure converts a text file into graphics output that can be displayed or printed on a graphics output device. You can enhance the output with TITLE, NOTE, and FOOTNOTE statements, or include Annotate graphics, or both. Like output from any other SAS/GRAPH procedure, output from the GPRINT procedure can be stored in catalogs and replayed with the GREPLAY procedure.

You can use the GPRINT procedure when you want to create graphics output from tabular material, reports, or any external text file produced by the SAS System or other software application. To display text and graphics generated by SAS/GRAPH software, use the GSLIDE procedure.

Figure 22.1 on page 859 shows a graphics output generated by the GPRINT procedure from SAS output generated by the MEANS procedure. Titles and footnotes have been added, and the Swiss font has been assigned to the procedure output text.

Figure 22.1 Graph Generated with the GPRINT Procedure

Regional Sales Report	
Region	Staff
NE	4
NW	3
SE	2
SW	4

Concepts

About External Text Files

External text files are files that you have stored outside of SAS. They can be created in several different ways. Four common methods are as follows:

- save the contents of the OUTPUT or LOG window to an external file with the FILE command
- direct the output from SAS procedures to an external file using the PRINTTO procedure and a FILENAME statement
- direct the output from a SAS data step to an external file using the FILE and PUT statements
- create a text file from another software application such as a text editor or a spreadsheet program.

Note: Depending on the operating environment and the method used to generate the file, external text files may contain carriage-control characters. For more information on carriage-control characters, see the NOCC option on page 862. Δ

You can use a FILENAME statement or host command to specify a fileref that points to the location of the external text file that you want to print. This external file serves as the input file for the GPRINT procedure.

Procedure Syntax

Global statements: FOOTNOTE, TITLE

Reminder: The procedure can include the NOTE statement.

Supports: Output Delivery System (ODS)

```
PROC GPRINT FILEREF=fileref  
  < option(s)>;
```

PROC GPRINT Statement

The PROC GPRINT statement identifies the external file to be converted to graphics output. Optionally, specifies the text color, a destination catalog for graphics output, and an Annotate data set.

Syntax

```
PROC GPRINT FILEREF=fileref  
  < option(s)>;
```

option(s) can be one or more of the following:

ANNOTATE=*Annotate-data-set*
 CTEXT=*text-color*
 DESCRIPTION=*'entry-description'*
 GOUT=< *libref.*>*output-catalog*
 NAME=*'entry-name'*
 NOCC
 O

Required Arguments

FILEREF=*fileref*

specifies the fileref that is associated with the external file that will be used as input to the GPRINT procedure. *Fileref* must have been previously defined in a FILENAME statement or host command.

See also: “FILENAME Statement” on page 24

Featured in: Example 2 on page 868

Options

Options in the PROC GPRINT statement affect all graphs that the statement produces. You can specify as many options as you want and list them in any order.

ANNOTATE=*Annotate-data-set*

ANNO=*Annotate-data-set*

specifies a data set to annotate the output that the GPRINT procedure produces.

See also: Chapter 10, “The Annotate Data Set,” on page 403

CTEXT=*text-color*

specifies the color in which the procedure displays the text from the input file.

If you do not use the CTEXT= option, a color specification is searched for in the following order:

- 1 the CTEXT= option in a GOPTIONS statement
- 2 the default, the first color in the colors list.

The CTEXT= option in the PROC GPRINT statement does not affect titles and footnotes generated by TITLE and FOOTNOTE definitions.

Featured in: Example 1 on page 865

DESCRIPTION=*'entry-description'*

DES=*'entry-description'*

specifies the description of the catalog entry for the chart. The maximum length for *entry-description* is 40 characters. The description does not appear on the chart. By default, the GPRINT procedure assigns the description OUTPUT FROM PROC GPRINT.

GOUT=< *libref.*>*output-catalog*

specifies the SAS catalog in which to save the graphics output produced by the GPRINT procedure. If you omit the libref, SAS/GRAPH looks for the catalog in the temporary library called WORK and creates the catalog if it does not exist.

See also: “Storing Graphics Output in SAS Catalogs” on page 49

NAME=*'entry-name'*

specifies the name of the catalog entry for the graph. The maximum length for *entry-name* is 8 characters. The default name is GPRINT. If the specified name duplicates the name of an existing entry, SAS/GRAPH software adds a number to the duplicate name to create a unique entry, for example, GPRINT1.

NOCC

tells the procedure that the external text file does not contain carriage-control characters. If you include the NOCC option, the procedure assumes that the first character on each line of the input file is a text character and not a carriage-control character. If you omit the NOCC option, the characters in column one are read as carriage-control characters. If they are valid carriage-control characters, the GPRINT procedure recognizes and executes them. If they are not valid carriage-control characters, the GPRINT procedure issues an error message.

O

causes a 0 (numeric zero) to be converted to the letter O in the output. This option circumvents the use of a numeric zero with an interior slash that is present on some devices.

Adjusting SAS Output and Graphics Output


The size of SAS output (or other text) in columns and rows and the size of graphics output are independently controlled. Depending on the result you want, you can do either of the following:

- Adjust the size of your SAS output (or other text) to fit the available space on your graph.
- Adjust the dimensions of the graphics output area and the size of the cells within the graphics output area to control the size of the characters that are displayed as graphics output by the GPRINT procedure.

You can adjust the size (columns and rows) of any other external text file that you use as input to the GPRINT procedure. Although the following sections explain how to adjust the size of SAS output, the general process can be applied to any text file.

SAS Output Size SAS output prints in pages. The length (in number of rows) and the width (in number of columns) of the page are determined by the PAGESIZE= and LINESIZE= options, respectively. Each character of SAS output occupies one column of space in a row (one cell), as shown in Figure 22.2 on page 862.

Figure 22.2 SAS Output Size



Graphics Output Size Graphics output is drawn in the graphics output area, which is also divided into cells. The overall dimensions of the graphics output area (width and height) are determined by the values of the device parameters XMAX and YMAX. These values, which determine the aspect ratio of the graphics output area, can be temporarily reduced with the HSIZE= and VSIZE= graphics options.

The number of columns and rows that fill the area is determined by the values of the LCOLS or PCOLS and LROWS or PROWS device parameters. These values, which

determine the size and aspect ratio of a cell, can be temporarily altered with the HPOS= and VPOS= graphics options. The more columns and rows there are in a given area, the smaller the cells are. Therefore, using HPOS= and VPOS= to change the number of columns and rows also changes the size of the cells and may change the size of the characters. However, it does not affect the overall dimensions of the graph. For details, see “Maintaining the aspect ratio of cells” on page 864.

See “Procedure Output and the Graphics Output Area” on page 29 for a complete description of the graphics output area. See Chapter 2, “SAS/GRAPH Programs,” on page 21 for more information on device parameters and graphics options.

Matching Sizes When you use the GPRINT procedure to convert SAS output to graphics output, you may need to manipulate the dimensions of either or both to get the proper size characters in the graphics output and to avoid truncating lines. Adjustment may be necessary in the following situations:

- If the number of rows per page in the SAS output (PAGESIZE=) exceeds the number of rows in the graphics output area (LROWS or PROWS), then the GPRINT procedure produces additional pages of graphics output.
- If the number of rows per page in the SAS output (PAGESIZE=) is much less than the number of rows in the graphics output area (LROWS or PROWS), then the output does not fill the graphics output area.
- If the width of a line of SAS output (LINESIZE=) exceeds the number of columns in the graphics output area (LCOLS or PCOLS), then the GPRINT procedure truncates the line.
- If the width of SAS output (LINESIZE=) is much less than the number of columns in the graphics output area (LCOLS or PCOLS), then the output does not fill the graphics output area.

You can adjust the size of the SAS output or the size of the graphics output, or both.

Adjusting the size of the SAS output The following steps show you how to use the PAGESIZE= and LINESIZE= options to adjust the page size of the SAS output to fit the size of the graphics output area.

- 1 Use the GDEVICE procedure to determine the number of rows (LROWS or PROWS) and the number of columns (LCOLS or PCOLS) on the graphics device that you intend to use. See Chapter 15, “The GDEVICE Procedure,” on page 651 for details.
- 2 Determine the number of columns and rows that you are going to use for SAS/GRAPH titles and footnotes. (If you specify height in units of CELLS, each unit of height equals one row.)
- 3 Use the OPTIONS statement to set the PAGESIZE= option equal to the number of rows on the device minus the number of positions to be used by TITLE and FOOTNOTE definitions. Set the LINESIZE= option equal to the number of columns on the device minus the number of positions used by titles and footnotes if the titles and footnotes are positioned vertically.
- 4 Produce the SAS output.

Adjusting the size of the graphics output The following steps show you how to use the HPOS= and VPOS= graphics options to adjust the number of columns and rows in the graphics output area on the output device so that it can accommodate the page size of your SAS output.

- 1 Determine the number of columns (LINESIZE=) and rows (PAGESIZE=) in the SAS output.
- 2 Use the GOPTIONS statement to set the VPOS= graphics option equal to the number of rows in the SAS output plus the number of rows to be used by TITLE

and FOOTNOTE definitions. Set the HPOS= graphics option equal to the number of columns in the SAS output plus the number of columns to be used by titles and footnotes if the titles and footnotes are positioned vertically.

3 Produce the GPRINT output.

Similarly, adjusting the overall dimensions of the graphics output area with the HSIZE= and VSIZE= graphics options may affect the size and possibly the aspect ratio of the cells.

Note: Changing the values of the HPOS= and VPOS= graphics options changes the size of the cells and consequently of characters in the output. On devices with nonscalable hardware fonts, changing the aspect ratio with HPOS= and VPOS= causes the Simulate font to be used instead of hardware characters. However, if you specify software fonts, the change in aspect ratio may be ignored. See “Using Fonts” on page 864 and “Using Hardware Fonts” on page 128 for more information. Δ

Maintaining the aspect ratio of cells If you change the values of the HPOS= and VPOS= graphics options to control the size of characters or to match the rows and columns of the external text file, you should try to maintain the same ratio of columns to rows as the original values of the device parameters. For example, suppose you have SAS output with 50 columns and 10 rows, and a graphics device that has 80 columns and 32 rows. The aspect ratio of the device is 5:2. If you print 10 rows of output on a device with 32 rows, you will have 22 blank lines. You can reduce the number of blank lines and increase the size of the characters by reducing the number of rows in the graphics output area with VPOS=. If, in addition to the 10 rows of output, you allow four lines of space for titles and two lines of space for a footnote, you need a total of 16 rows. Therefore, assigning a value of 20 to VPOS= should produce readable text and plenty of space. If VPOS=20, setting HPOS= to 50 retains the original aspect ratio of the device (80:32 or 5:2).

Note that this method allows space for titles and footnotes in terms of rows; the actual size of the titles and footnotes depends on the height specification you use. Using the unit CELLS to define the height of titles and footnotes makes it easier to calculate precisely how much space is available.

Using Fonts

By default, the GPRINT procedure uses the default hardware font with a height of 1 cell to display the text from the external file. However, if you specify a nonscalable hardware font, SAS/GRAPH may use the Simulate font instead. See Chapter 6, “SAS/GRAPH Fonts,” on page 125 for details.

Font and height specifications for titles and footnotes are determined by the TITLE and FOOTNOTE definitions. See “TITLE, FOOTNOTE, and NOTE Statements” on page 251 for details.

To specify a font and height for the text, use the FTEXT= and HTEXT= graphics options. If you specify a software font, it is best to use a uniform font such as Swiss Uniform so that your text will be evenly spaced.

CAUTION:

Changes in the aspect ratio of cells made with the HPOS= and VPOS= graphics options are ignored if you specify software fonts. Change the aspect ratio in the device entry if you want the software characters proportioned to fit the new aspect ratio. Δ

If you specify a software font and change the aspect ratio of the cells with the HPOS= and VPOS= graphics options, the change in aspect ratio is ignored and the procedure continues to draw the font in the original proportions. As a result, your text may not fit the graphics output area.

However, if you want the software characters to reflect a change in aspect ratio or if you want the characters to fit the new aspect ratio even if they are distorted, use the LCOLS or PCOLS and LROWS or PROWS device parameters in the device entry to change the aspect ratio of the cells. Using the device entry to specify a change in the aspect ratio enables you to distort the characters. See Chapter 15, “The GDEVICE Procedure,” on page 651 for more information on changing device parameters.

Examples

Example 1: Specifying Color Text

Procedure features:

GPRINT procedure options:

CTEXT=

Other features:

GOPTIONS statement


TIMEPLOT procedure

Sample library member: GR22N01

This example creates the REFLIB.DOWHLC data set and generates a graph with color text from output that is produced by the TIMEPLOT procedure. The TIMEPLOT procedure is not a graphics procedure and produces text output only. (See *SAS Procedures Guide* for details on the TIMEPLOT procedure.)


The first part of this example uses the TIMEPLOT procedure with the newly created REFLIB.DOWHLC data set as input to produce Output 22.1 on page 865:

Output 22.1 SAS Output from the TIMEPLOT Procedure



The second part of this example takes the output generated by the TIMEPLOT procedure and converts it to a graph by using the GPRINT procedure. Figure 22.3 on page 866 shows the graph with color text, a title, and a footnote:

Figure 22.3 GPRINT Procedure Output with Enhanced Text (GR22N01)



Assign the libref and set the graphics environment. HTEXT= assigns the height for the text in the default unit, cells.

```
libname reflib 'SAS-data-library';
options reset=global border cback=white
        colors=(black blue green red)
        ftitle=swissb htitle=3pct
        htext=.8 ftext=none
        hsize=7in vsize=5in;
```

Assign the fileref OUT to the external file.

```
filename out 'external-file';
```

Create the data set REFLIB.DOWHLC.

```
data reflib.dowhlc;
  input date date9. high low close;
  format date date9.;
  datalines;
02JAN1997    6511.38    6318.96    6442.49
03JAN1997    6586.42    6437.10    6544.09
...more data lines...
30JAN1997    6845.03    6719.96    6823.86
31JAN1997    6912.37    6769.99    6813.09
;
```

Suppress the date line and page numbers and set the linesize and pagesize.

```
options nodate nonumber linesize=80 pagesize=60;
```

Specify the destination for all subsequent procedure output.

```
proc printto print=out new;
run;
```

Generate TIMEPLOT graph output. It is sent to external file.

```
proc timeplot data=reflib.dowhlc;
  plot low close high / overlay hiloc ref=mean(low)
                                npp axis=6310 to 6960 by 10;
  id date;
run;
```

Reset destination for printed output to default.

```
proc printto;
run;
```

Define title and footnote.

```

title 'TIMEPLOT of Dow-Jones Averages';
footnote h=3 pct f=swiss
        j=l ' L=Low' ' C=Close' ' H=High'
        j=r 'GR22N01 ' ;

```

Generate graph from the external file and specify text color. CTEXT= assigns a color to the text produced by the GPRINT procedure.

```

proc gprint fileref=out ctext=red;
run;

```

Example 2: Adjusting the Size of Characters

Procedure features:

GPRINT statement options:

FILEREF=

Other features:

FILENAME statement

GOPTIONS statement

PRINT procedure

PRINTTO procedure

Data set: REFLIB.DOWHLC on page 867

Sample library member: GR22N02

This example creates a graph from a text file and increases the size of the text. The first part of this example uses the PRINT procedure to create an external file that contains SAS output. The GPRINT procedure is used to import the text file into a graph. Because the LINESIZE= option (columns) is set to 76 and the PAGESIZE= option (rows) is set to 24, the output is small and occupies only a portion of the page, as shown in Figure 22.4 on page 869:

Figure 22.4 GPRINT Procedure Output with No Adjustments (GR22N02(a))

Dow-Jones Averages				
Obs	date	high	low	close
1	02JAN1997	6511.38	6318.96	6442.49
2	03JAN1997	6586.42	6437.10	6544.09
3	06JAN1997	6647.22	6508.30	6567.18
4	07JAN1997	6621.82	6481.75	6600.66
5	08JAN1997	6650.30	6509.84	6549.48
6	09JAN1997	6677.24	6520.23	6625.67
7	10JAN1997	6725.35	6530.62	6703.79
8	13JAN1997	6773.45	6647.99	6709.18
9	14JAN1997	6816.17	6689.94	6762.29
10	15JAN1997	6800.77	6669.93	6726.88
11	16JAN1997	6818.47	6688.40	6765.37
12	17JAN1997	6863.88	6732.66	6833.10
13	20JAN1997	6839.13	6777.30	6843.87
14	21JAN1997	6934.69	6771.14	6883.90
15	22JAN1997	6913.14	6801.16	6850.03
16	23JAN1997	6953.55	6724.19	6755.75
17	24JAN1997	6798.08	6629.91	6696.48
18	27JAN1997	6748.82	6598.73	6660.69
19	28JAN1997	6823.48	6612.20	6656.08
20	29JAN1997	6673.39	6627.98	6740.74
21	30JAN1997	6845.03	6719.96	6823.86
22	31JAN1997	6912.37	6769.99	6813.09

GR22N02(a)

In the second part of this example, the number of columns and rows in the graphics output area is reduced with the HPOS= and VPOS= graphic options. Thus, the size of the characters in the graph increase, as shown in Figure 22.5 on page 869:

Figure 22.5 GPRINT Procedure Output with Adjusted Sizing (GR22N02(b))

Dow-Jones Averages				
Obs	date	high	low	close
1	02JAN1997	6511.38	6318.96	6442.49
2	03JAN1997	6586.42	6437.10	6544.09
3	06JAN1997	6647.22	6508.30	6567.18
4	07JAN1997	6621.82	6481.75	6600.66
5	08JAN1997	6650.30	6509.84	6549.48
6	09JAN1997	6677.24	6520.23	6625.67
7	10JAN1997	6725.35	6530.62	6703.79
8	13JAN1997	6773.45	6647.99	6709.18
9	14JAN1997	6816.17	6689.94	6762.29
10	15JAN1997	6800.77	6669.93	6726.88
11	16JAN1997	6818.47	6688.40	6765.37
12	17JAN1997	6863.88	6732.66	6833.10
13	20JAN1997	6839.13	6777.30	6843.87
14	21JAN1997	6934.69	6771.14	6883.90
15	22JAN1997	6913.14	6801.16	6850.03
16	23JAN1997	6953.55	6724.19	6755.75
17	24JAN1997	6798.08	6629.91	6696.48
18	27JAN1997	6748.82	6598.73	6660.69
19	28JAN1997	6823.48	6612.20	6656.08
20	29JAN1997	6673.39	6627.98	6740.74
21	30JAN1997	6845.03	6719.96	6823.86
22	31JAN1997	6912.37	6769.99	6813.09

GR22N02(b)

Assign the libref and set the graphics environment. FTEXT= in the GOPTIONS statement specifies the default hardware font. (This is the default setting.)

```
libname reflib 'SAS-data-library';
goptions reset=global border cback=white
          colors=(black blue green red)
          ftitle=swissb ftext=none
          hsize=7in vsize=5in
```

```
hpos=142 vpos=68;
```

Assign the fileref DOW to the external file. The fileref DOW is associated with the external file where the output from PROC PRINT is stored.

```
filename dow 'external-file';
```

Suppress the date line and page numbers. Set the line and page size.

```
options nodate nonumber linesize=76 pagesize=24;
```

Specify the destination for all subsequent procedure output. The PRINTTO procedure directs the SAS output to the external file that the GPRINT procedure subsequently uses as input. PRINT= directs all printed procedure output to the file referenced by the fileref DOW. NEW causes the output file to be replaced each time the program is run.

```
proc printto print=dow new;
run;
```

Send the output to the destination file. The PRINT procedure generates the text and sends it to the external file specified by PROC PRINTTO.

```
proc print data=reflib.dowhlc;
run;
```

Reset destination for printed output to the default. The destination for printed output is reset to the default by resubmitting PROC PRINTTO with no options.

```
proc printto;
run;
```

Define title and footnote.

```
title 'Dow-Jones Averages';
footnote h=3 pct f=swiss j=r 'GR22N02(a) ';
```

Generate graph from the external file. FILEREF= specifies the external file that is used as input. NOCC is omitted because the input text file contains carriage-control characters.

```
proc gprint fileref=dow;
run;
```

Reduce HPOS= and VPOS= to increase cell size.

```
goptions hpos=75 vpos=30;
```

Define the footnote.

```
footnote h=3 pct f=swiss j=r 'GR22N02(b) ';
```

Generate adjusted graph.

```
proc gprint fileref=dow;  
run;
```


The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/GRAPH® Software: Reference, Version 8*, Cary, NC: SAS Institute Inc., 1999.

SAS/GRAPH® Software: Reference, Version 8

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-525-6

All rights reserved. Printed in the United States of America.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

OS/2®, OS/390®, and IBM® are registered trademarks or trademarks of International Business Machines Corporation.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.